

13 Multiple Integrals

13.1 Integration with More Than One Variable

Let R be a closed region in the plane and f a function defined on R . In this section we will discuss how Maple V can be used to study two-variable integrals such as

$$\int_R f(x, y) dA$$

In particular, let the region R be the rectangle:

$$\{(x, y) : a \leq x \leq b, c \leq y \leq d\}.$$

Then the integral is defined to be

$$\int_R f(x, y) dA = \lim_{n \rightarrow \infty} \lim_{m \rightarrow \infty} \sum_{j=0}^{m-1} \left(\sum_{i=0}^{n-1} f(a + i \Delta x, b + j \Delta y) \Delta x \right) \Delta y,$$

where

$$\Delta x = \frac{b-a}{n}, \Delta y = \frac{d-c}{m}.$$

We will now illustrate this with a specific example. Let's approximate the double integral

$$\int_R 7 - x^2 - y dA$$

where R is the rectangle

$$\{(x, y) : 0 \leq x \leq 2, 2 \leq y \leq 3\}.$$

In other words we wish to approximate the volume of the region lying above R and under the graph of

$$z := f(x, y) = 7 - x^2 - y$$

First we define the function.

```
> f := (x,y) -> 7 - x^2 - y;
```

$$f := (x, y) \rightarrow 7 - x^2 - y$$

Next we plot the solid whose volume we wish to approximate. See Figure 57.

```
> plot3d(f(x,y), x=0..2, y=2..3, axes=BOXED, tickmarks=[3,3,3], style=PATCH);
```

Before we begin let's review the Maple V command **sum**. For example:



```
> r := sum(1/n^2, n=1..infinity);
```

$$r := \frac{1}{6} \pi^2$$

This is the sum of $\frac{1}{n^2}$ with n having the values 1, 2, 3, ... all the way up to ∞ .

For our first approximation of the double integral let's use 625 subrectangles by selecting m and n each to be 25. Execute the following commands.

```
> a:=0;b:=2;c:=2;d:=3;
```

Figure 57: Plot of $z = 7 - x^2 - y^2$ over $[0, 2] \times [2, 3]$ Figure 58: Plot of $z = (x^2 + y^2)^{1/2}$

```

a := 0

b := 2

c := 2

d := 3

> m:=25;n:=25;

m := 25

n := 25

> delx := (b-a)/m; dely := (d-c)/n;

delx :=  $\frac{2}{25}$ 

dely :=  $\frac{1}{25}$ 

> approxvol:=sum(sum(f(a+i*delx,c+j*dely)*delx,i=0..m-1)*dely,j=0..n-1);

approxvol :=  $\frac{4082}{625}$ 

> evalf(approxvol);

```

6.531200000

Let's see how much of a change there is if we select m and n to be 100 so that there will be 10000 subrectangles.

```
> m := 100; n:= 100;
```

$$m := 100$$

$$n := 100$$

```
> delx:= (b-a)/m; dely := (d-c)/n;
```

$$\text{delx} := \frac{1}{50}$$

$$\text{dely} := \frac{1}{100}$$

```
> approxvol:=sum(sum(f(a+i*delx,c+j*dely)*delx,i=0..m-1)*dely,j=0..n-1);
```

$$\text{approxvol} := \frac{7979}{1250}$$

```
> evalf(approxvol);
```

6.383200000

There is a noticeable change in the value we obtained. The more rectangles the region is divided into for summation, the more accurate the result. Let's go wild and let m and n be 10000. This produces 100,000,000 subrectangles.

```
> m := 10000; n:= 10000;
```

$$m := 10000$$

$$n := 10000$$

```
> delx:=(b-a)/m;dely:=(d-c)/n;
```

$$\text{delx} := \frac{1}{5000}$$

$$\text{dely} := \frac{1}{10000}$$

```
> approxvol:=sum(sum(f(a+i*delx,c+j*dely)*delx,i=0..m-1)*dely,j=0..n-1);
```

$$\text{approxvol} := \frac{158345833}{25000000}$$

```
> evalf(approxvol);
```

6.333833320

Note that the value changed again, but not by as much as before. Perhaps we can feel confident that at least the first two digits of our approximation are correct. Let's see what Maple V returns for a value if we evaluate the ITERATED integral:

$$\int_0^2 \int_2^3 7 - x^2 - y \, dy \, dx$$

First we integrate the “inside” integral.

```
> insideint := int(f(x,y), y=2..3);
```

$$insideint := \frac{9}{2} - x^2$$

Now we integrate this with respect to x .

```
> int(insideint, x=0..2);
```

$$\frac{19}{3}$$

This is the EXACT value of the volume. We see that the approximation was pretty good, since

```
> evalf(19/3);
```

6.333333333

The value obtained by summation with 100,000,000 subrectangles was close but by no means exact. In most cases it is easier and more accurate to use the built-in Maple V command **int** to calculate integrals, but ever now and then you'll run across an example that even Maple V can't do. If we are only wanting the answer to this problem we could have accomplished that with the following command.

```
> int(int(f(x,y), y=2..3), x=0..2);
```

$$\frac{19}{3}$$

Sometimes even Maple V can't evaluate an iterated integral in an obvious way. For example, consider

$$\int_0^1 \int_x^1 e^{(y^2)} \, dy \, dx$$

Let's do two integrations separately so that we can follow the Maple V steps.

```
> insideint := int(exp(y^2), y=x..1);
```

$$insideint := -\frac{1}{2} I \sqrt{\pi} \operatorname{erf}(I) + \frac{1}{2} I \sqrt{\pi} \operatorname{erf}(I x)$$

This seems like a strange answer, but Maple V can complete the answer anyway.

```
> int(insideint, x=0..1);
```

$$\frac{1}{2} e - \frac{1}{2}$$

Had we issued the command

```
> int(int(exp(y^2), y=x..1), x=0..1);
```

$$\frac{1}{2}e - \frac{1}{2}$$

we would have missed the excitement that went on with the inside integral.

If we had to work the problem by hand, we could make the integration much more straightforward by reversing the order of integration, producing the iterated integral

$$\int_0^1 \int_0^y e^{(y^2)} dx dy$$

```
> insideint:=int(exp(y^2),x=0..y);
```

$$insideint := e^{(y^2)} y$$

```
> int(insideint,y=0..1);
```

$$\frac{1}{2}e - \frac{1}{2}$$

There are times when Maple V has the same problem that we have in trying to perform the integration in the "given" order. For example, consider the iterated integral

$$\int_0^2 \int_{x^2}^4 x^3 \sin(y^3)^2 dy dx$$

If you try to have Maple V compute this integral, you will have to wait a long time to find out that Maple V can't do it.

```
> int(int(x^3*sin(y^3)^2,y=x^2..4),x=0..2);
```

$$\int_0^2 \int_{x^2}^4 x^3 \sin(y^3)^2 dy dx$$

However, reversing the order of integration works.

```
> insideint:=int(x^3*sin(y^3)^2,x=0..sqrt(y));
```

$$insideint := \frac{1}{4} y^2 \sin(y^3)^2$$

```
> ANS := int(insideint,y=0..4);
```

$$ANS := -\frac{1}{24} \cos(64) \sin(64) + \frac{8}{3}$$

If we wish to have a 10 digit approximation to this exact value, then

```
> evalf(ANS);
```

2.651645048

As before the combined command

```
> int(int(x^3*sin(y^3)^2,x=0..sqrt(y)),y=0..4);
```

would have produced the same result. But remember Maple V has its limits, just like we do. There are times when reversing the order of integration will not enable Maple V to solve a particular problem. When this happens, you either give up or turn to a numerical approximation scheme such as Riemann sums.

As example of a triple iterated integral let us consider the following problem:
Compute the mass of the cone bounded by

$$z = \sqrt{x^2 + y^2}$$

and $z = 3$, if the density is given by $\rho(x, y, z) = z$.

We will plot the cone. See Figure 58.

```
> plot3d(sqrt(x^2+y^2),x=-3..3,y=-sqrt(9-x^2)..sqrt(9-x^2),
> style=PATCHNOGRID,axes=BOXED,tickmarks=[3,3,0],orientation=[20,65]);
```

The iterated integral which is equal to the mass is

$$\int_{-3}^3 \int_{-\sqrt{9-x^2}}^{\sqrt{9-x^2}} \int_{\sqrt{x^2+y^2}}^3 z \, dz \, dy \, dx$$

The following Maple V command gives the exact answer.

```
> int(int(int(z,z=sqrt(x^2+y^2)..3),y=-sqrt(9-x^2)..sqrt(9-x^2)),x=-3..3);
```

$$\frac{81}{4} \pi$$

Exercises 13.1:

1. Estimate the double integral

$$\int_R (4x^3 + 6xy^2) \, dA,$$

where

$$R = [1, 3] \times [-2, 1],$$

using $100^2=10,000$ subrectangles. See how close your answer is to the exact value by calculating an appropriate iterated double integral exactly.

2. Evaluate the double iterated integral

$$\int_{y^2}^{y+2} \int_{-1}^2 \sqrt{z-y^2} \, dz \, dy.$$

3. Evaluate the double iterated integral

$$\int_0^1 \int_0^1 \sqrt{4-x^2-y^2} \, dx \, dy.$$

13.2 Using the Monte Carlo Method to Approximate Integrals Numerically

As was mentioned in the preceding section, there are integration problems for which even Maple V can not find an elementary antiderivative. Consequently numerical methods must be utilized. In this section we use a scheme called the Monte Carlo method. Consider the unit circle as inside the square with vertices $(-1, -1)$, $(1, -1)$, $(1, 1)$, $(-1, 1)$. See Figure 59. Imagine randomly throwing darts at this figure. We would expect that some darts would land in the square outside the circle and some would land inside the circle. It seems reasonable to guess, that after many dart tosses, the ratio of the number of darts landing inside the circle, N_R , to the total number, N , of darts thrown gives an estimate of the ratio of the area of the circle to the area of the square. Since the area of the square is 4 an estimate for the area of the circle would be given by $4 \frac{N_R}{N}$.




Figure 59: Unit Circle Inside Square

Maple V has a built-in procedure **rand** which allows us to simulate the random tossing of darts. There are a number of different syntaxes for using this procedure, but here we will call the routine with one argument. The command **rand(Q)** where Q is a positive integer returns a procedure that when called outputs an integer which is (pseudo-) randomly chosen from the set of integers $\{0, 1, \dots, Q - 1\}$. The following Maple V segments illustrates the method.

First we produce two functions, x and y , which randomly generates a pair inside the given square.

```
> x := evalf(rand(10001)/5000-1):
```

```
> y := evalf(rand(10001)/5000-1):
```

Next we prepare to make 1000 tosses at the square and count the number of times the dart lands in the circle:

```
> N := 1000;
```

```
N := 1000
```

```
> NR := 0;
```

```
NR := 0
```

```
> for i from 1 to N do
```

```
> if x()^2 + y()^2 < 1 then NR := NR +1 fi
```

```
> od:
```

Since the ratio $\frac{N_R}{N}$ should approximate the ratio of the area of the circle to the area of the square, and since the area of the square is 4, we conclude that four times this ratio should be an approximation for the area of the circle.

```
> 4*evalf(NR/N);
3.152000000
```

The last Maple V output should be an approximation for the area of the unit circle which is π . So the approximation is not so good, but if we use larger values for N , we might expect better results. In any case, this gives us an estimate of the double integral

$$\int_R dx dy,$$

where R is the unit circle.

As another example, consider approximating the double integral

$$\int_0^1 \int_0^1 e^{-(x^2+y^2)} dx dy$$

by the Monte Carlo method.

This integral gives the volume of the solid lying above the unit square $[0, 1] \times [0, 1]$ and under the surface $z = e^{-(x^2+y^2)}$. Since this solid lies within the unit cube $[0, 1] \times [0, 1] \times [0, 1]$ and the unit cube has volume 1 the ratio, $\frac{N_R}{N}$, should be an estimate for the volume. In this example we add to the count N_R whenever the randomly generated point (x, y, z) which lies in the cube satisfies

$$0 \leq z \leq e^{-(x^2+y^2)}.$$

The following Maple V segment illustrates how the Monte Carlo method can be implemented with 1000 trials.

```
> x := evalf(rand(10000)/9999);
> y := evalf(rand(10000)/9999);
> z := evalf(rand(10000)/9999);
> N := 1000;
> NR := 0;

> for i from 1 to N do
> if z() < exp(-x()^2-y()^2) then NR := NR +1 fi
> od:

> evalf(NR/N);
.5580000000
```

We conclude that

$$\int_0^1 \int_0^1 e^{-(x^2+y^2)} dx dy \approx .5580000000$$

As a comparison let us use Maple V to estimate the value of this integral:

```
> FIRST := int(exp(-t^2-s^2), t=0..1);
FIRST :=  $\frac{\sqrt{\pi} \operatorname{erf}(1)}{2e^{s^2}}$ 
> evalf(int(FIRST, s=0..1));
```


$$0.5577462855$$

It follows that

$$\int_0^1 \int_0^1 e^{-(x^2+y^2)} dx dy \approx 0.5577462855$$

so the result obtained by the Monte Carlo method gave a pretty good estimate.

Exercises 13.2 Apply the Monte Carlo Method to estimate the value of the following integrals. Compare your answers with values for the integrals obtained by another method.

1.

$$\int_0^1 \int_0^{\sqrt{1-x^2}} \sqrt{1-x^2-y^2} dy dx$$

2.

$$\int_0^1 \int_0^1 xy^{xy} dx dy$$

3.

$$\int_0^1 \int_0^1 \sqrt{4-x^2-y^2} dx dy.$$

13.3 Two-Variable Integrals in Polar Coordinates

In this section we illustrate how Maple V can be used to compute certain integration problems using polar coordinates:

$$x = r \cos(\theta), y = r \sin(\theta)$$

Suppose that you are asked to find the following integral

$$\int_R \frac{1}{(x^2 + y^2)^{3/2}} dA$$

over the pie-shaped annular region R which lies in the first quadrant bounded between the two circles:

$$x^2 + y^2 = 1, x^2 + y^2 = 4$$

and the lines $y = 0$ and $y = x$.

As a first step plot the region R using Maple V and rectangular coordinates. The region R is plotted by the following command. See Figure 60.

```
> with(plots):
> P1 := plot(sqrt(1-x^2), x=1/sqrt(2)..1): #The arc of the circle x^2 + y^2 = 1
> P2 := plot(sqrt(4-x^2), x=sqrt(2)..2): # The arc of circle x^2+y^2 = 4
> P3 := plot(x, x=1/sqrt(2)..sqrt(2)): # The line y = x
> display({P1,P2,P3}, view = [0..2, 0..2], scaling=constrained);
```




Figure 60: Region between $x^2 + y^2 = 1$, $x^2 + y^2 = 4$, $y = 0$, $y = x$




Figure 61: Transformed Region in (r, θ) Space

Can you set up a double iterated integral to evaluate this double integral? It is quite a difficult task that requires breaking the region R up into various subregions. However, the integral becomes quite simple in polar coordinates. We first plot it using Maple V and polar coordinates. This is another way to produce Figure 60.

```
> P1 := plot([2,t,t=0..Pi/4], 0..2, 0..2, coords=polar): # Circle r = 2
> P2 := plot([1,t,t=0..Pi/4], 0..2, 0..2, coords=polar): # Circle r = 1
> P3 := plot([t,Pi/4,t=1..2], 0..2, 0..2, coords=polar): # Line y = x
> display({P1,P2,P3}, scaling=constrained);
```

In polar coordinates the integral is the following:

```
> I1 := Int(Int((1/r^3)*r, r=1..2), theta=0..Pi/4);
```

$$I1 := \int_0^{1/4\pi} \int_1^2 \frac{1}{r^2} dr d\theta$$

```
I1 := value(I1);
```

$$I1 := \frac{1}{8} \pi$$

```
> int(int(1/r^2,r=1..2),theta=0..Pi/4);
```

$$\frac{1}{8} \pi$$

It is interesting to observe that if we plot the region R using the very same **plot** command but omitting the **coords = polar** we obtain a rectangular region which provides the limits of integration for the problem. See Figure 61

```
> P1 := plot([2,t,t=0..Pi/4],0..2,0..2):
```

```
> P2 := plot([1,t,t=0..Pi/4],0..2,0..2):
```

```
> P3 := plot([t,Pi/4,t=1..2],0..2,0..2):
```

```
> display({P1,P2,P3},scaling=constrained);
```

The integral of a function $F(r, \theta)$ over this region is

$$\int_0^{1/4\pi} \int_1^2 F(r, \theta) dr d\theta$$

which is the same as the integral used for polar coordinates. There is also a way to convert an integrand of the form $f(x, y)dx dy$ to its appropriate form when using a coordinate change $x = x(u, v)$, $y = y(u, v)$ by using the Jacobian. The rule is that the integrand in the u, v coordinates becomes:

$$f(x(u, v), y(u, v)) \left| \frac{\partial(x(u, v), y(u, v))}{\partial u \partial v} \right| du dv.$$

For example, in the problem that we just did the integrand using rectangular coordinates is

$$\frac{dx}{dy} (x^2 + y^2)^{3/2}$$

Lets use the Jacobian to determine the integrand using polar coordinates:

$$x = r \cos(\theta), y = r \sin(\theta)$$

and Maple V.

First we call up the linear algebra package so that we can use the procedures **jacobian** and **det**.

```
> with(linalg):
```

```
Warning: new definition for norm
```

```
Warning: new definition for trace
```

Now we compute value of the determinant of the jacobian matrix. (We will omit the absolute value function in the following, since all the terms turn out to be positive.)

```
> dA := det(jacobian([r*cos(theta), r*sin(theta)], [r, theta]))*dr*dtheta;
```

$$dA := |\cos(\theta)^2 r + r \sin(\theta)^2| dr d\theta$$

```
> dA := simplify(dA);
```

$$dA := |r| dr d\theta$$

```
> F := subs(x=r*cos(theta), y=r*sin(theta), 1/(x^2+y^2)^(3/2));
```

$$F := \frac{1}{(r^2 \cos(\theta)^2 + r^2 \sin(\theta)^2)^{3/2}}$$

```
> F := simplify(F);
```

$$F := \frac{1}{(r^2)^{3/2}}$$

```
> simplify(expand(F)*dA, symbolic);
```

$$\frac{dr d\theta}{r^2}$$

The last result gives the integrand that was used in this example.

Example: Convert the iterated integral

$$\int_{-1}^0 \int_{-\sqrt{1-x^2}}^{\sqrt{1-x^2}} f(x, y) dy dx$$

to polar coordinates.

Solution: The region R over which this integration takes place is the half disk inside the unit circle and to the left of the y axis. See Figure ??.

```
> plot([1,t,t=Pi/2..3*Pi/2],[t,Pi/2,t=0..1],[t,3*Pi/2,t=0..1]),
```

```
> coords = polar, scaling = constrained,
```


```
> xtickmarks=2, ytickmarks=2);
```

If we omit the option **coords = polar** we get a rectangular region in the r, θ plane that provides the limits of integration. See Figure ??.

```
> plot([1,t,t=Pi/2..3*Pi/2],[t,Pi/2,t=0..1],[t,3*Pi/2,t=0..1]),
```

```
> 0..1, 0..3*Pi/2,
```

```
> scaling = constrained, xtickmarks=2, ytickmarks=2);
```

Figure 62: Unit Half Disk $x^2 + y^2 = 1, y \geq 0$ Figure 63: Transformed Region in (r, θ) Space

Now replacing $dx dy$ with the jacobian expression we complete the problem:

```
> Int(Int(subs(x=r*cos(theta),y=r*sin(theta),f(x,y))*det(jacobian(
> [r*cos(theta),r*sin(theta)], [r,theta])),r=0..1),theta=Pi/2..3*Pi/2);
```

$$\int_{1/2\pi}^{3/2\pi} \int_0^1 f(r \cos(\theta), r \sin(\theta)) (\cos(\theta)^2 r + r \sin(\theta)^2) dr d\theta$$

```
> simplify(");
```

$$\int_{1/2\pi}^{3/2\pi} \int_0^1 f(r \cos(\theta), r \sin(\theta)) r dr d\theta$$

Example: Let's consider the problem of finding the area to the right of the vertical line $x = 7/8$ that is inside the circle

$$r = 3 \sin(\theta)$$

Solution: First we will plot this region and in the process use the command **polarplot** which is part of the **plots** package.

```
> with(plots):
The next command plots the circle.
> P1 := polarplot(3*sin(theta),theta=0..Pi,scaling=constrained):";
The vertical line  $x = 7/8$  has equation
> r = 7/8*sec(theta);
\l
\{r\}={\displaystyle \frac {7}{8}}\,\{\rm sec\}(\,\{\ \theta\}\,,)
\]
\end{maplelatex}
```

Thus we can use `\bf polarplot` to plot the line.

```
\begin{maple}
> P2 := polarplot(7/8*sec(theta),theta=0..5*Pi/12,scaling=constrained):";
```

We need to find the area of the region to the right of the line which lies inside the circle. See Figure ??.

```
> display({P1,P2});
```




Figure 64: Circle $x^2 + y^2 = 9$, and Line $x = 7/8$

In order to do this we need values of θ where the line and circle intersect. Say these values are

θ_1 , and, θ_2

respectively, then the area is obtained from the following double integral.

$$\int_{\theta_1}^{\theta_2} \int_{7/8 \sec(\theta)}^{3 \sin(\theta)} r \, dr \, d\theta$$

In order to find the points of intersection study the following plot. Since we plan to use **fsolve** to find the points of intersection we need to specify the intervals in which Maple V is to seek a solution.

```
> plot({7/8*sec(theta), 3*sin(theta)}, theta=0..5*Pi/12);
```

This suggests that the intersections occur in the interval's $[0.2, 0.4]$, and $[1, 1.3]$. We now use **fsolve** to estimate these points to 10 digits.

```
> theta[1] := fsolve(7/8*sec(x)=3*sin(x), x, 0.2..0.4);
```

$\theta_1 := .3114132927$

```
> theta[2] := fsolve(7/8*sec(x)=3*sin(x), x, 1..1.3);
```

$\theta_2 := 1.259383034$

Finally, the following double integral should give a very good approximation to the area.

```
> Int(Int(r, r=7/8*sec(theta)..3*sin(theta)), theta=theta[1]..theta[2]);
```

$$\int_{.3114132927}^{1.259383034} \int_{7/8 \sec(\theta)}^{3 \sin(\theta)} r \, dr \, d\theta$$

```
> value(");
```

1.066876285

We conclude that the area of the region is around 1.066876285.

Exercise 13.3

1. Evaluate the double integral

$$\int_R y^2 dA$$

where R is the region in the first quadrant that is outside the circle $r = 2$ and inside the cardioid $r = 2(1 + \cos \theta)$.

2. Find the volume V of the solid above the polar rectangle

$$R = \{(r, \theta) | 1 \leq r \leq 3, 0 \leq \theta \leq \frac{\pi}{4}\}$$

and under the surface $z = x e^{(x^2+y^2)/\pi}$.

3. Evaluate

$$\int_0^1 \int_0^1 \sqrt{4 - x^2 - y^2} dx dy.$$

13.4 Multiple Integrals in Various Coordinates

Let W be a region in three space and let f be defined on W . If it turns out that W can be defined in terms of rectangular coordinates, then the integral over the region can usually be evaluated by means of a triple iterated integral like the following:

$$\int_W f(x, y, z) dV = \int_a^b \int_{U(z)}^{V(z)} \int_{u(y,z)}^{v(y,z)} f(x, y, z) dx dy dz$$

As an example let's plot the region W which is the intersection of two right circular cylinders of radius one. One is symmetric about the x -axis and the other symmetric about the z -axis. You should convince yourself that the following Maple V commands plot the cylinders which are shown in Figure ??.

```
> W1 := plot3d([x,cos(theta),sin(theta)],theta=0..2*Pi,x=-1..1,
> scaling =constrained,style=wireframe):";
> W2 := plot3d([cos(theta),sin(theta),z],theta=0..2*Pi,z=-1..1,
> scaling =constrained,style=wireframe):";
> with(plots):
> display3d({W1,W2});
```




Figure 65: Intersection of Cylinders




Figure 66: Intersection of Cylinders on Smaller Domain

Let us find the volume of the solid W formed by the intersection of these two cylinders. Note that W is made up of eight identical regions so we can concentrate on the part which lies in the first octant. Now take a look at a picture of this piece. See Figure ??

```
> W3 := plot3d([x,cos(theta),sin(theta)],theta=0..Pi/2,x=0..1,
> scaling =constrained,style=wireframe):";
> W4 := plot3d([cos(theta),sin(theta),z],theta=0..Pi/2,z=0..1,
> scaling =constrained,style=wireframe):";
> display3d({W3,W4});
The volume of this piece is
> PieceVolume:=Int(Int(Int(1,z=0..sqrt(1-y^2)),x=0..sqrt(1-y^2)),
> y=0..1);
```


$$PieceVolume := \int_0^1 \int_0^{\sqrt{1-y^2}} \int_0^{\sqrt{1-y^2}} 1 \, dz \, dx \, dy$$

```
> PieceVolume := value(PieceVolume);
```

$$PieceVolume := \frac{2}{3}$$

and thus the volume of the entire solid is eight times that of the piece we just calculated.

```
> TotalVolume := 8*PieceVolume;
```

$$TotalVolume := \frac{16}{3}$$

We used a triple integral to find the volume above, but we could also have used a double integral:

```
> PieceVolume := Int(Int(sqrt(1-y^2), x=0..sqrt(1-y^2)), y=0..1);
```

$$PieceVolume := \int_0^1 \int_0^{\sqrt{1-y^2}} \sqrt{1-y^2} \, dx \, dy$$

```
> PieceVolume := value(PieceVolume);
```

$$PieceVolume := \frac{2}{3}$$

Now let's do another problem with this same solid. Let T be the top half of the solid W , and suppose the density of T at any point (x, y, z) is equal to the square of the distance from the point to the plane $z = 0$, i.e.,

```
> rho(x,y,z) = z^2;
```

$$\rho(x, y, z) = z^2$$

```
> Tmass := 4*Int(Int(Int(z^2, z=0..sqrt(1-x^2)), y=0..sqrt(1-x^2)),
> x=0..1);
```

$$Tmass := 4 \int_0^1 \int_0^{\sqrt{1-y^2}} \int_0^{\sqrt{1-y^2}} z^2 \, dz \, dx \, dy$$

```
> Tmass := value(Tmass);
```

$$Tmass := \frac{32}{45}$$

Now to find the center of mass of T you must find the moment of T about the plane $z = 0$.

```
> Tmomentxy := 4*Int(Int(Int(z^3, z=0..sqrt(1-y^2)), x=0..sqrt(1-y^2)),
> y=0..1);
```

$$Tmomentxy := 4 \int_0^1 \int_0^{\sqrt{1-y^2}} \int_0^{\sqrt{1-y^2}} z^3 \, dz \, dx \, dy$$

```
> Tmomentxy := value(Tmomentxy);
```

$$Tmomentxy := \frac{5}{32} \pi$$

```
> zbar := Tmomentxy/Tmass;
```

$$zbar := \frac{225}{1024} \pi$$

Thus the center of mass of T is located at $(0, 0, 225\pi/1024)$.

Just as it is sometimes convenient to use polar coordinates when computing double integrals over appropriate regions, one can sometimes use generalizations of polar coordinates to aid in calculating triple integrals.

The relationship between cylindrical coordinates and rectangular coordinates is

$$x = r \cos(\theta), y = r \sin(\theta), z = z$$

When changing coordinates one must determine what to use for dV . This is done using the Jacobian.

```
> with(linalg):
```

```
Warning: new definition for norm
Warning: new definition for trace
```

```
> JacobianC := jacobian([r*cos(theta), r*sin(theta), z], [r, theta, z]);
```

$$JacobianC := \begin{bmatrix} \cos(\theta) & -r \sin(\theta) & 0 \\ \sin(\theta) & r \cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

along with the determinant

```
> DetjacobianC := simplify(det(JacobianC));
```

$$DetjacobianC := r$$

This means that when evaluating triple integrals with cylindrical coordinates that we use

$$dV = r dr d\theta dz$$

As an example consider the solid W formed below the plane $z = 1$ and above the upper half of the right circular cone



$$z = \sqrt{x^2 + y^2}$$

Lets plot this cone using **cylinderplot**. See Figure ??.

```
> with(plots):
> cylinderplot(z, theta = 0..2*Pi, z=0..1);
The volume of  $W$  using cylindrical coordinates is
> Volume := Int(Int(Int(r, z=r..1), r=0..1), theta=0..2*Pi);
```

$$Volume := \int_0^{2\pi} \int_0^1 \int_r^1 r dz dr d\theta$$

```
> Volume := value(Volume);
```

Figure 67: **Cylinderplot** of a ConeFigure 68: **Cylinderplot** of Paraboloid

$$\text{Volume} := \frac{1}{3} \pi$$

As another example showing how cylindrical coordinates can be used to evaluate triple integrals let us convert the following triple integral in rectangular coordinates to one in cylindrical coordinates.

```
> Int ( Int ( Int ( x^2+y^2, z=0..4-x^2-y^2 ), y=-sqrt(4-x^2)..sqrt(4-x^2) ),
> z=-2..2 );
```

$$\int_{-2}^2 \int_{-\sqrt{4-x^2}}^{\sqrt{4-x^2}} \int_0^{4-x^2-y^2} x^2 + y^2 \, dz \, dy \, dx$$

This is a triple integral of the function $f(x, y, z) = x^2 + y^2$ over the solid W bounded above by the paraboloid $z = 4 - x^2 - y^2$, $z \geq 0$, and bounded below by the plane $z = 0$.

Let's plot this region with **cylinderplot**. See Figure ??.

```
> cylinderplot(sqrt(4-z), theta=0..2*Pi, z=0..4, scaling=constrained);
```

The integral in cylindrical coordinates becomes

```
> V := Int ( Int ( Int ( r^2*r, z=0..4-r^2 ), r=0..2 ), theta=0..2*Pi );
```

$$V := \int_0^{2\pi} \int_0^2 \int_0^{4-r^2} r^3 \, dz \, dr \, d\theta$$

```
> V:= value(V);
```

$$V := \frac{32}{3} \pi$$

Spherical coordinates are related to rectangular coordinates as follows:

$$x = \rho \sin(\phi) \cos(\theta), y = \rho \sin(\phi) \sin(\theta), z = \rho \cos(\phi)$$

Using the determinant of the Jacobian of this transformation we obtain an expression for dV .

```
> JacobianS :=jacobian(
> [rho*sin(phi)*cos(theta), rho*sin(phi)*sin(theta),
> rho*cos(phi)], [rho, phi, theta]);
```

$$JacobianS := \begin{bmatrix} \sin(\phi) \cos(\theta) & \rho \cos(\phi) \cos(\theta) & -\rho \sin(\phi) \sin(\theta) \\ \sin(\phi) \sin(\theta) & \rho \cos(\phi) \sin(\theta) & \rho \sin(\phi) \cos(\theta) \\ \cos(\phi) & -\rho \sin(\phi) & 0 \end{bmatrix}$$

```
> DetJacobianS := simplify(det(JacobianS));
```

$$DetJacobianS := \sin(\phi) \rho^2$$

Thus in spherical coordinates we have

$$dV = \rho^2 \sin(\phi) d\rho d\theta d\phi$$

As an example consider the solid bounded below by the upper half of the cone

$$\phi = \frac{1}{3}\pi$$

and above by the sphere

$$\rho = 1$$

First we plot this region using **sphereplot** and **cylinderplot**. The sphere is plotted by

```
> P1 := sphereplot(1, theta = 0..2*Pi, phi=0..Pi/2):";
\end{mapleinput}
```

As before we obtain the cone as follows:

```
\begin{maple}
> P2 := cylinderplot(z, theta=0..2*Pi, z=0..1):";
```

The multiple plot shown in Figure ?? is given by next command:

```
> display3d({P1,P2}, style = wireframe, axes=normal);
```

The volume of this solid is given by the following triple integral:

```
> V := Int(Int(Int(rho^2*sin(phi), rho=0..1), phi=0..Pi/3),
> theta=0..2*Pi);
```

$$V := \int_0^{2\pi} \int_0^{1/3\pi} \int_0^1 \sin(\phi) \rho^2 d\rho d\phi d\theta$$

```
> V := value(V);
```

$$V := \frac{1}{3}\pi$$

Exercises 13.4

1. Find the volume of the three dimensional solid enclosed by the surfaces

$$z = x^2 + 4y^2, \text{ and } z = 12 - 2x^2 - 2y^2.$$

2. Find the mass of the solid bounded by the planes $y = 0$ and $z = 0$ and by the surfaces $z = 4 - x^2$ and $x = y^2$ assuming that the density function is $\delta(x, y, z) = xy^2z^3$.

3. Find the volume cut from the sphere $x^2 + y^2 + z^2 = 9$ by the cone $z = 3\sqrt{x^2 + y^2}$.

4. Find the volume of the region enclosed by the cylinder $x^2 + y^2 = 16$ and the planes $z = 0$ and $z + y = 16$.




Figure 69: Plot of Cone Inside Sphere